
Historically, all IBES data have been adjusted for stock splits. The Summary History data, however, 

did not have enough precision to allow researchers to unadjust the data without experiencing severe 

rounding errors
1
. To accommodate the needs of researchers that need to use IBES data that are not 

adjusted for stock splits, IBES now makes several files available in unadjusted form.  

 

The files that are available in unadjusted form are: Detail History - Estimates, Detail History - 

Actuals, Detail History - Excluded Estimates, Summary History - Summary Statistics, and Summary 

History - Actuals, Pricing & Ancillary. The WRDS SAS data set names for these files are: detu, 

actu, excu, statsumu and actpsumu, respectively. 

 

IBES only provides unadjusted data for the EPS measure. No data for the other measures are 

available in unadjusted form at this time. 

 

Merging Unadjusted Actuals data with Unadjusted Detail History and Summary History data is 

problematic, and for this reason WRDS has elected so far not to merge the actuals data with the 

other files. A detailed discussion of this issue follows. 

 

Care must be taken when merging unadjusted data files to make sure that aligned EPS data items are 

based on the same number of shares outstanding (e.g., EPS estimates and EPS actuals). Problems 

can arise in the vicinity of a split date if one data item accounts for the split but the other does not. 

For example, if a split occurs between an analyst's Estimate Date and the associated Report Date of 

Earnings, the estimates and actual values may be based on different numbers of shares outstanding. 

Under these circumstances, simply merging the Detail History Estimates data with the Detail History 

Actuals data by matching Ticker, the Period End Date and periodicity variables (annual, quarterly, 

semin-annual) will result in mismatches like the following: 

 
TICKER ESTDATS BROKER ANALYS VALUE FPEDATS REVDATS REPDATS Actual 

AMZN 5/20/1998 86 42186 -2.5 12/31/1998 6/3/1998 1/26/1999 -0.517 

AMZN 5/26/1998 192 1830 -2.1 12/31/1998 7/23/1998 1/26/1999 -0.517 

AMZN 5/28/1998 229 32051 -3.06 12/31/1998 6/3/1998 1/26/1999 -0.517 

AMZN 6/4/1998 86 42186 -1.17 12/31/1998 7/22/1998 1/26/1999 -0.517 

AMZN 6/4/1998 229 32051 -1.12 12/31/1998 7/23/1998 1/26/1999 -0.517 

AMZN 6/8/1998 191 45029 -1.21 12/31/1998 6/9/1998 1/26/1999 -0.517 

…………………………………………………………………………………………………………………………… 

AMZN 1/5/1999 241 259 -0.67 12/31/1998 6/23/2000 1/26/1999 -0.517 

AMZN 1/5/1999 100 30593 -0.54 12/31/1998 6/26/2000 1/26/1999 -0.517 

AMZN 1/5/1999 229 32051 -0.56 12/31/1998 6/26/2000 1/26/1999 -0.517 

AMZN 1/14/1999 899 53564 -0.54 12/31/1998 6/26/2000 1/26/1999 -0.517 

 

 

Amazon split 2-for-1 on Jun 2, 1998 and 3-for-1 on Jan 5, 1999. When you compare the above 

annual EPS estimates (“Value”) to the actual reported values (“Actual”), it appears that estimates 

made in May 1998 are roughly 6 times lower than the actual value, estimates made in June 1998 are 

about 2 times lower than the actual reported value and estimates made on or after Jan 5, 1999 appear 

to be on the same order of magnitude as the actual value. These mismatches occur because 

unadjusted analyst estimates are based on the number of shares outstanding as of the Estimate Date, 

while the actual reported earnings are based on the number of shares outstanding as of the Report 

                                                 
1
 For more on the issue of rounding errors in IBES adjusted data, see Payne, Jeff L. and Wayne B, Thomas (2003). "The 

Implications of Using Stock-Split Adjusted I/B/E/S Data in Empirical Research.". Accounting Review, 78(4), 1049-1067. 

Date of Earnings (the date when the company announced their actual EPS for the fiscal period – in 

this case, Jan. 26, 1999). Because Amazon stock split between these two dates, the estimates and 

actuals are sometimes based on different numbers of shares outstanding. This implies that surprises 

based on these estimates and actuals will be wrong in cases when there is a mismatch. 

 

There are three possible ways one can think of to address this mismatch problem.  

 

Method 1 

 

The first method is based on adjusting actual values using IBES adjustment factor valid as of the 

report date and then unadjusting this adjusted actual using IBES adjustment factor valid as of the 

estimate date. Namely, 

 

1. Merge the Adjustment Factors file into the Actuals file by matching the Effective Split Date 

with the Report Date and then adjust the actual values for splits; 

 

2. Merge the adjusted Actuals data with the unadjusted Estimates data (Detail History or 

Summary History), matching on the Period End Date and periodicity variables; 

 

3. Merge the Adjustment Factors into the Estimates / Actuals data by matching the Effective 

Split Date with either the Estimate Date (for Detail History) or the Statistical Period (for 

Summary History); 

 

4. Unadjust the actuals by applying the contemporaneous split factor as of the Estimate Date 

(Detail History) or Statistical Period (Summary History) date. 

 

This technique puts most - but not all - Estimate / Actual pairs on the same per-share basis. For 

example, the results for the above sample of Amazon EPS estimates are: 

 
TICKER ESTDATS BROKER ANALYS VALUE FPEDATS REVDATS REPDATS Actual 

AMZN 5/20/1998 86 42186 -2.5 12/31/1998 6/3/1998 1/26/1999 -3.102 

AMZN 5/26/1998 192 1830 -2.1 12/31/1998 7/23/1998 1/26/1999 -3.102 

AMZN 5/28/1998 229 32051 -3.06 12/31/1998 6/3/1998 1/26/1999 -3.102 

AMZN 6/4/1998 86 42186 -1.17 12/31/1998 7/22/1998 1/26/1999 -3.102 

AMZN 6/4/1998 229 32051 -1.12 12/31/1998 7/23/1998 1/26/1999 -3.102 

AMZN 6/8/1998 191 45029 -1.21 12/31/1998 6/9/1998 1/26/1999 -3.102 

………………………………………………………………………………………………………………………… 

AMZN 1/5/1999 241 259 -0.67 12/31/1998 6/23/2000 1/26/1999 -1.551 

AMZN 1/5/1999 100 30593 -0.54 12/31/1998 6/26/2000 1/26/1999 -1.551 

AMZN 1/5/1999 229 32051 -0.56 12/31/1998 6/26/2000 1/26/1999 -1.551 

AMZN 1/14/1999 899 53564 -0.54 12/31/1998 6/26/2000 1/26/1999 -0.517 

 

Notice that the estimates made in May 1998 now appear to be on the same per-share basis as the 

actual. The same is true for estimate on Jan 14, 1999. There still appears to be a problem with the 

Jun 1998 estimate / actual pairs and there is a newly created mismatch with estimate/actual pairs 

corresponding to Jan 5, 1999.  

The reason for this discrepancy is that the IBES Effective Split Date is not necessarily the true date 

of the stock split. Instead, it is the date when the split became effective within the IBES database. 

Below is the extract from the IBES adjustment factor file: 

 

 
    TICKER     CUSIP      OFTIC         CNAME           SPDATES    ADJ    SPDATES 

 

     AMZN     02313510    AMZN     AMAZON.COM INC.    18JUN1998     6     980618       

 

     AMZN     02313510    AMZN     AMAZON.COM INC.    14JAN1999     2     990114       

 

     AMZN     02313510    AMZN     AMAZON.COM INC.    16SEP1999     1     990916     

 

IBES Effective Split Date usually occurs around the middle of the month (as does the IBES 

Statistical Period in the Summary History file), and is almost never the same as the true split date. 

Recall, that the true split dates for Amazon were Jun 2, 1998 and Jan 5, 1999, NOT Jun 18, 1998 and 

Jan 14, 1999. Analyst estimates, on the other hand, are based on the true number of shares 

outstanding as of the Estimate Date, since analysts know exactly when a stock splits.  

 

Because step (3.) of the above procedure merges the adjustment factors into the Estimates file by 

matching the Effective Split Date with the Estimate Date, mismatches can occur when an estimate is 

made between the true split date and the Effective Split Date (for example, there was an estimate 

made on Jun 4, 1998, right between the true split date of Jun 2, 1998 and IBES effective Split Date 

of Jun 18, 1998). In these cases the estimates include the effect of the recent split while the actual 

does not. The fact that IBES Effective Split date is not a true split date also results in occurrence of 

additional mismatches (in Amazon example, this happened for estimates made on Jan 5, 1999). 

 

When this technique is used to merge the Summary Statistics and Actuals files, however, the results 

are significantly better. This is due to the fact that in this case, step (3.) matches the Effective Split 

Date with the IBES Statistical Period, and both of these are mid-month dates when certain things 

become "effective" within the IBES database. But although the results of the merge appear to be 

very good, it is still possible to find exceptions. And once again, the source of the problem is that the 

IBES Effective Split Date is not the same as the true split date. The following example illustrates the 

problem: 

 

Ticker    Per. End Date   Stat. Period    Report Date     Mean    Actual 

MSFT    31DEC1989    14DEC1989    18JAN1990      1.00     1.2600 

MSFT    31DEC1989    18JAN1990     18JAN1990      1.01     1.2600 

MSFT    31MAR1990   15FEB1990     17APR1990      1.16     0.6187 

MSFT    31MAR1990   15MAR1990    17APR1990     1.16     0.6187 

MSFT    30JUN1990     19APR1990     25JUL1990      0.59     0.6502 

MSFT    30JUN1990     17MAY1990    25JUL1990     0.62     0.6502 

 

These Summary History data are quarterly, 1-period-ahead mean EPS estimates ("Mean") and actual 

values ("Actual") for Microsoft. It appears that the mean estimate / actual value pairs for the periods 

ending in Dec., 1989 and June, 1990 are on the same per-share basis, but it also appears that the 

estimate / actual pairs for the period ending in Mar., 1990 are not. The source of this problem can be 

traced to step (1.) of our procedure, in which the Adjustment Factors file is merged into the Actuals 

file by matching the Effective Split Date with the Report Date. Once again, a problem arises because 

the true split date is not necessarily the same as the Effective Split Date. Whenever the Report Date 

of earnings comes between the true date of a stock split and the Effective Split Date, this problem 

can occur. 

 

Microsoft stock split 2-for-1 on April 12, 1990, they reported earnings (for the quarter ending 

March, 1990) on April 17, and the Effective Split Date is April 19. Since the actual value is based on 

the number of shares outstanding as of the Report Date and the Report Date comes after the true split 

date, the actual value includes the effect of the split. On the other hand, the Mean estimate is based 

on the number of shares outstanding relative to the Effective Split Date, so it does not include the 

effect of the split. Although this problem is relatively rare – it requires that the true split date < 

Report Date < Effective Split Date, and this relationship rarely holds – it is another example of the 

problems that can arise when merging IBES unadjusted data files and the true split date is unknown. 

 

Method 2 

Whenever IBES needs to apply a split factor to their data, they use SPDATES. But analysts are not 

constrained in this way. They know exactly when a stock split and base their estimates on the 

number of shares outstanding as of the Estimate Date (ESTDATS). This reasoning suggests one 

more way to correct mismatch in estimates/actual pairs.  

One can get the correct split factor for each estimate by merging the unadjusted data with the 

adjusted data, and then backing out the split factor. This makes use of the analyst's information 

regarding the split date rather than IBES' dubious SPDATES variable. Namely, this procedure 

implies the following algorithm: 

1. Merge the Adjustment Factors file into the Actuals file by matching the Effective Split Date 

with the Report Date and then adjust the actual values for splits; 

 

2. Merge the adjusted Actuals data with the unadjusted Estimates data (Detail History or 

Summary History), matching on the Period End Date and periodicity variables; 

 

3. Merge the Adjusted Detail data from Ibes.Det (set A) into the Estimates / Actuals data (set B) 

obtained in step 2 , for example, using the following  

 
A.ticker=B.ticker and  

    A.fpedats=B.fpedats and  

    A.measure=B.measure and  

 A.usfirm=B.usfirm and  

 A.estdats=B.estdats and  

 A.analys=B.analys and  

 A.broker=B.broker and  

 A.fpi=B.fpi and  

 A.revdats=B.revdats 
  

and back out the implied adjustment factor by dividing unadjusted estimate from ibes.detu by 

its adjusted counterpart from ibes.det. 

 

4. Unadjust the adjusted actuals from step 1 by applying the implied split factor which should 

be valid as of the Estimate Date (Detail History) or Statistical Period (Summary History) 

date. 

In our previous example with Amazon, this results in the following matching: 

 

TICKER ESTDATS BROKER ANALYS VALUE FPEDATS REVDATS REPDATS Actual 

AMZN 5/20/1998 86 42186 -2.5 12/31/1998 6/3/1998 1/26/1999 -3.102 

AMZN 5/26/1998 192 1830 -2.1 12/31/1998 7/23/1998 1/26/1999 -3.102 

AMZN 5/28/1998 229 32051 -3.06 12/31/1998 6/3/1998 1/26/1999 -3.102 

AMZN 6/4/1998 86 42186 -1.17 12/31/1998 7/22/1998 1/26/1999 -1.551 

AMZN 6/4/1998 229 32051 -1.12 12/31/1998 7/23/1998 1/26/1999 -1.551 

AMZN 6/8/1998 191 45029 -1.21 12/31/1998 6/9/1998 1/26/1999 -1.551 

AMZN 1/5/1999 241 259 -0.67 12/31/1998 6/23/2000 1/26/1999 -0.517 

AMZN 1/5/1999 100 30593 -0.54 12/31/1998 6/26/2000 1/26/1999 -0.517 

AMZN 1/5/1999 229 32051 -0.56 12/31/1998 6/26/2000 1/26/1999 -0.517 

AMZN 1/14/1999 899 53564 -0.54 12/31/1998 6/26/2000 1/26/1999 -0.517 

Note that now all estimate/actual pair values appear to be on the same per-share basis. However, 

discrepancies may still occur due to the rounding effects in IBES Detail Adjusted History files. 

Another problem with this approach is that the number of observations in IBES Adjusted Detail 

History does not match the number of observations in IBES Unadjusted Detail History. This implies 

that the implied adjustment factor may not be obtainable for all EPS estimates in the unadjusted 

dataset.  

Method 3  

Perhaps, relatively the most reliable and accurate way of joining Unadjusted Detail History and 

Actuals is to use CRSP cumulative adjustment split factor extracted from the CRSP Daily files as it 

contains more precise information regarding the true split date of a stock compared with IBES split 

factor . This method involves the following steps: 

1. Merge the unadjusted Estimates data (Detail History or Summary History) with unadjusted 

Actuals data matching on the Period End Date and periodicity variables. 

2. Merge the resulting dataset with IBES-CRSP linking table (which can be obtained by 

running /wrds/ibes/samples/iclink.sas) and select a list of unique Permnos from the merged 

dataset. 

3. Extract a subset from CRSP Daily file which contains Permno, Date and Cumulative Share 

Adjustment Factor by doing inner join with Permnos obtained from step 2  

4. Merge dataset from step 2 with CRSP daily file extract from step 3 by matching on Permno 

and Estimate Date. This will give adjustment factor valid as of estimate date 

5. Merge dataset from set 4 with CRSP daily file extract from step 3 by matching on Permno 

and Report Date. This will give adjustment factor valid as of report date 

6. Compute the correct actual value by multiplying unadjusted IBES actual by the ratio of 

cumulative adjustment factor as of estimate date to that of report date. 

 

The procedure described above results in the following matching between unadjusted estimates and 

actuals: 

 

 

 

TICKER ESTDATS BROKER ANALYS VALUE FPEDATS REVDATS REPDATS Actual CumAdj_est CumAdj_rep

AMZN 5/20/1998 86 42186 -2.5 12/31/1998 6/3/1998 1/26/1999 -3.102 12 2

AMZN 5/26/1998 192 1830 -2.1 12/31/1998 7/23/1998 1/26/1999 -3.102 12 2

AMZN 5/28/1998 229 32051 -3.06 12/31/1998 6/3/1998 1/26/1999 -3.102 12 2

AMZN 6/4/1998 86 42186 -1.17 12/31/1998 7/22/1998 1/26/1999 -1.551 6 2

AMZN 6/4/1998 229 32051 -1.12 12/31/1998 7/23/1998 1/26/1999 -1.551 6 2

AMZN 6/8/1998 191 45029 -1.21 12/31/1998 6/9/1998 1/26/1999 -1.551 6 2

AMZN 1/5/1999 241 259 -0.67 12/31/1998 6/23/2000 1/26/1999 -0.517 2 2

AMZN 1/5/1999 100 30593 -0.54 12/31/1998 6/26/2000 1/26/1999 -0.517 2 2

CumAdj_est and CumAdj_rep are the CRSP cumulative adjustment factors valid as of Estimate and 

Report date respectively. They correctly capture the stock’s split history: between May, 1998 

(estimate month in first three rows) and Jan 26, 1999 (corresponding report date) Amazon stock split 

twice, once 2-for-1 and one more time 3-for-1, so the ratio CumAdj_est/CumAdj_rep=6, after the 

stock split on June 2, 1998 the ratio becomes 3. So multiplying IBES unadjusted actual by this ratio 

allows us to put the estimate and actual on the same per-share basis. 

 

Researchers may find the techniques described here to be a useful starting point for their own, more 

sophisticated merging technique. Therefore, although WRDS has not merged the Unadjusted Actuals 

into the Unadjusted Detail History and Summary History files for the reasons described above, the 

SAS programs that perform this technique can be found in the appendix below and are made 

available on the WRDS Unix server at: 

 

/wrds/ibes/samples/unadjust_merge.sas 

 

WRDS subscribers are welcome to modify and use these programs at their own risk. Comments and 

suggestions are always welcome via e-mail to wrds-support@wharton.upenn.edu 

Appendix. Sample Code 

 
/* ***************************************************************************** */ 

/* Program      :      unadjust_merge.sas                                        */  

/* Author       :      David Robinson and Denys Glushkov, WRDS                   */ 

/* Version      :      1.0                                                       */ 

/* Date Created :   November 10, 2006                                         */  

/* Description  :   Merge unadjusted actuals and ID data into Detail History  */ 

/*                  Unadjusted data. For more detailed description of the    */ 

/*      purpose of this program, see ‘A Note on IBES Unadjusted   */ 

/*      Data’ which can be found in ‘Quick Links’ section of IBES */ 

/*      homepage             */  

/*                                                                               */  

/* Note:                                                                         */ 

/* 1. The programs "actu.sas", "adj.sas", "detu1.sas" and "id.sas" should        */ 

/* be run before this program.                                                   */ 

/* 2. You need to have permission on both IBES data and CRSP Daily Stock data    */ 

/* to run this program                                                           */ 

/*                     */  

/* Output: Creates and Downloads a SAS file ‘unadjust_merge’ into SAS Work     */ 

/*   directory on your desktop, which contains analyst forecasts and actuals */ 

/*   from IBES unadjusted file which are on the same per-share basis     */ 

/* ----------------------------------------------------------------------------- */ 

/*                                                                               */ 

/* Unadjusted EPS data are stated relative to the number of shares               */ 

/* outstanding as of a certain date. Therefore, each observation in              */ 

/* every table must have a date variable that is compared to the                 */ 

/* effective split date to determine the basis (shares outstanding)              */ 

/* of the value. If two tables contain the same date variable but                */ 

/* one uses that variable to set shares outstanding but the other                */ 

/* table uses a different date variable, it is likely that many                  */ 

/* estimate / actual value comparisons will not be valid because                 */ 

/* the basis (shares outstanding) of the two values will be                      */ 

/* different. The Unadjusted Actuals table compares the Period End               */ 

/* Date to the Effective Split Date to set shares outstanding,                   */ 

/* while the Detail History Unadjusted table compares ESTDATS /                  */ 

/* REVDATS to the split date. Therefore, merging the Actuals with                */ 

/* the Detail History Estimates by matching Period End Dates will                */ 

/* not result in valid Estimate / Actual comparisons in all cases.               */ 

/* ----------------------------------------------------------------------------  */ 

/*                                                                               */ 

/*  The following program suggests three different ways to address               */ 

/*  this problem, with the most time-consuming and resource-intensive third      */ 

/*  method being, probably, relatively most accurate and reliable                */ 

/* ----------------------------------------------------------------------------- */ 

/*                                                                               */ 

/*  First method one is using the following algorithm:                           */ 

/*  1. Merge the Adjustment Factors file into the Actuals file by matching       */ 

/*  the Effective Split Date with the Report Date and then adjust                */ 

/*  the actual values for splits;                                                */  

/*  2. Merge the adjusted Actuals data with the unadjusted Estimates data        */ 

/*  (Detail History or Summary History), matching on the Period End Date         */ 

/*  and periodicity variables;                                                   */ 

/*  3. Merge the Adjustment Factors into the Estimates / Actuals data by         */ 

/*  matching the Effective Split Date with either the Estimate Date (for Detail  */ 

/*  History) or the Statistical Period (for Summary History);                    */ 

/*  4. Unadjust the actuals by applying the contemporaneous split factor         */ 

/*  as of the Estimate Date (Detail History) or Statistical                      */ 

/*  Period (Summary History) date.                                               */ 

/* ----------------------------------------------------------------------------- */ 

/*                                                                               */ 

/* Second method follows the similar methodology to the first method except      */ 

/* instead of steps 3 and 4 it merges the adjusted data with the unadjusted data */  

/* and then backs out the split factor, which is applied to the adjusted actuals */ 

/* obtained in step 1. The advantage of the approach is that it makes use        */  

/* of the analyst's information regarding the split date rather than IBES'       */  

/* dubious SPDATES variable                                                      */ 

/*                                                                               */ 

/* ----------------------------------------------------------------------------  */ 

/*                                                                               */ 

/*  Third method makes use of IBES-CRSP linking table (iclink.sas) to obtain     */ 

/*  CRSP cumulative split adjustment factors valid as of two different dates:    */ 

/*  analyst estimate date and report date. Then IBES actual unadjusted values are*/ 

/*  scaled by the ratio of these two adjustment factors to obtain actual value   */ 

/*  which is on the same per-share basis as an analyst estimate                  */ 

/*                                                                               */ 

/* ***************************************************************************** */ 

  

/* --------------------------------------------------------------- */ 

/* remote connect to WRDS Server using PC SAS CONNECT              */ 

/* Note:                                                           */ 

/*      if you use the program in WRDS UNIX server, please remove  */ 

/* the signon code (the next three lines) and all the rsubmit and  */ 

/* endrsubmit statements                                           */  

/* --------------------------------------------------------------- */ 

%let wrds = wrds.wharton.upenn.edu 4016; 

options comamid=TCP; 

signon wrds username=_prompt_; 

 

rsubmit; 

libname ibes '/wrds/ibes/sasdata'; 

 

options nosource; 

options source; 

options nocenter nodate ps=max ls=max mprint symbolgen macrogen msglevel=i; 

 

/* ------------------------------------- */ 

/* select tickers and time interval      */  

/* ------------------------------------- */ 

%let screen=(ticker in ("DELL", "IBM", "MSFT", "PG","JNJ","AMZN")); 

/* 

%let year1=1998; 

%let year2=2000; 

%let begdate = "01JAN&year1"d; 

%let enddate = "31DEC&year2"d; 

*/ 

/* ---------------------------------------------------------------  */ 

/* Select and sort data from the Unadjusted Detail History Actuals  */ 

/* file and define repweek as the TRADING (non-weekend)             */ 

/* date either equal to the report date if it is not weekend or     */ 

/* closest trading day preceeding the report date, if the latter is */ 

/* Saturday or Sunday. We will need this variable                   */ 

/* later to merge with CRSP split adjustmentr factors               */ 

/* ---------------------------------------------------------------  */ 



 9 

 

data actual; 

set ibes.actu (keep=ticker pends repdats pdicity value usfirm); 

where &screen /*and &begdate<=repdats<=&enddate*/; 

date = repdats; 

if weekday(repdats)=1 then repweek=repdats-2;else 

if weekday(repdats)=7 then repweek=repdats-1;else 

repweek=repdats; 

format date date9.;format repweek date9.; 

run; 

 

title 'Extract from Detail History Actuals Unadjusted'; 

proc print data=actual (obs=50 drop=usfirm date); 

run; 

 

proc sort data=actual; 

 by ticker date usfirm; 

run; 

 

/* ---------------------------------------------------------- */ 

/* Select and sort data from the IBES Adjustment Factors file */ 

/* ---------------------------------------------------------- */ 

data split ; 

set ibes.adj (keep=ticker spdates adj usfirm); 

where &screen /*and &begdate<=spdates<=&enddate*/; 

date = spdates; 

format date date9.; 

run; 

 

title 'Extract from Adjustment Factor File'; 

proc print data=split (obs=50); 

run; 

 

proc sort data=split; 

 by ticker date usfirm; 

run; 

 

data split; 

set split; 

by ticker date; 

if first.ticker then date = '01JAN1900'd; 

run; 

 

/* ------------------------------------------------------------- */ 

/* Merge the Adjustment Factors into the Actual Unadjusted data, */ 

/* create a Cumulative Adjustment Factor variable (CADJFAC), and */ 

/* create an adjusted actual value variable, ACTVAL_ADJ.         */ 

/* ------------------------------------------------------------- */ 

 

data act; 

merge actual (in=actdata) split (in=spldata) ; 

by ticker date usfirm; 

format CADJFAC 12.6 ACTVAL_ADJ 14.8; 

retain lsplit lspdates; 

if first.ticker then do; 

  if actdata and not spldata then CADJFAC = 1.0000; 

  else CADJFAC = adj; 

  spdates = .B; 

end; 

else do; 

  if actdata and not spldata then do; 

    CADJFAC = lsplit; 

    spdates = lspdates; 

  end; 

  else CADJFAC = adj; 

end; 

lsplit = CADJFAC; 

lspdates = spdates; 

actval_adj = value / cadjfac; 

drop date lsplit lspdates adj; 

if actdata then output; 

run; 

 

title 'Merged Actuals&Adjustment with adjusted actuals and cumulative adjustment factor'; 

proc print data=act (obs=50 keep=ticker spdates repdats value actval_adj cadjfac); 

run; 

 

/* -------------------------------------------------------- */ 

/* Create an adjustment factors file containing the unique  */ 

/* TICKER / SPDATES combinations (w/ CADJFAC).              */ 

/* -------------------------------------------------------- */ 

proc sort data=act out=adjfac (keep=ticker spdates cadjfac usfirm) nodupkey; 

 by ticker spdates usfirm; 

run; 

 

data adjfac; 

set adjfac; 

date=spdates; 

run; 

 

data detu1; 

set ibes.detu; 

where &screen; 

run; 

 

title 'Extract from Detail History Estimates'; 

proc print data=detu1 (obs=50 where=(fpi='1') 

    keep=ticker fpedats estdats analys value fpi); 

run; 

 

/* ------------------------------------------------- */ 

/* Merge the actual re-adjusted data into the detail */ 

/* history estimates file.                           */ 

/* ------------------------------------------------- */ 

proc sql; 

  create table det_unadj as  

  select detu1.*, act.actval_adj, act.repdats, actval_unadj, act.repweek, detu1.estdats as date 

  from detu1 (rename=(value=forecast)) left join act (rename=(value=actval_unadj)) 

  on  detu1.ticker=act.ticker  and  detu1.usfirm=act.usfirm and  

      detu1.measure ='EPS'   and  detu1.fpedats=act.pends and  

     ( ( detu1.fpi in ("1","2","3","4","5","E","F","G","H","I","X") and act.pdicity = "ANN")  

  or 

       ( detu1.fpi in ("6","7","8","9","N","O","P","Q","Y") and act.pdicity = "QTR")  

  or 

       ( detu1.fpi in ("A","B","C","D","Z") and act.pdicity = "SAN")  

     );   

quit; 

 

title 'Merged Detail History Estimates and Actuals unadjusted'; 

proc print data=det_unadj (obs=50 keep=ticker fpedats estdats forecast fpi 

                          actval_unadj actval_adj where=(fpi='1')); 

run; 

 

proc sort data=det_unadj; 

 by ticker date usfirm; 

run; 

/*-------------------------------------------------------------------*/ 

/* Extract a sample from the IBES adjusted Detail History to         */  

/* merge with unadjusted IBES Detail History and back out adjustment */ 

/* factor (second method)                                            */ 

/* ----------------------------------------------------------------- */ 

 

data det_adj;  

set ibes.det (keep=ticker fpedats measure value usfirm estdats broker revdats analys fpi); 

where &screen; 

rename value=forecast_adj; 

run; 

 

proc sql; create table  

   det_unadj  

as select * 

from det_unadj as a left join det_adj as b 

on  a.ticker=b.ticker and  

    a.fpedats=b.fpedats and  

    a.measure=b.measure and  

 a.usfirm=b.usfirm and  

 a.estdats=b.estdats and  

 a.analys=b.analys and  

 a.broker=b.broker and  

 a.fpi=b.fpi and  

 a.revdats=b.revdats 

order by ticker, date, usfirm; 

quit; 

 

title 'Merged Detail History Adjusted and Unadjusted'; 

proc print data=det_unadj (obs=50 keep=ticker fpedats estdats forecast forecast_adj   

                           actval_unadj actval_adj); 

run; 

 

/*------------------------------------------------------------------ */ 

/* Obtain IBES ticker - CRSP permno link using                       */ 

/* IBES-CRSP linking table                                           */ 

/*-------------------------------------------------------------------*/ 

 

%include '/wrdslink/ibes/samples/iclink.sas'; /*creates IBES Ticker - Permno Link*/ 

 

libname mylib './'; /* define your library here */ 

libname crsp '/wrds/crsp/sasdata/sd'; 

 

/*----------------------------------------------------------------- */ 

/* Get Permnos corresponding to IBES tickers                        */  

/*----------------------------------------------------------------- */ 

proc sql; create table det_unadj (drop=score) 

as select * 

from det_unadj as a left join mylib.iclink (keep=ticker permno score  

where=(score in (0,1,2))) as b 

on a.ticker=b.ticker; 

quit; 

 

proc sort data=det_unadj nodupkey out=permnos (keep=permno); 

 by permno; 

run; 

 

/* Extract daily cumulative adjustment factors for the list of permnos */  

/* and merge them into Detail History file                             */ 

/*-------------------------------------------------------------------- */ 

proc sql; create table crspadj 

as select a.* 

from crsp.dsf (keep=permno date CFACSHR where=(date>='01dec1974'd)) as a, permnos as b 

where a.permno=b.permno; 

quit; 

 

/*Get CRSP Cumulative Adjustment Factor as of Estimate Date*/ 

proc sql; create table det_unadj (drop=permno_date) 

as select * 

from det_unadj as a left join crspadj (rename=(CFACSHR=crspadj_est date=permno_date)) as b 

on a.permno=b.permno and a.estdats=b.permno_date; 

quit; 

 

/*Get CRSP Cumulative Adjustment Factor as of Report Date*/ 

proc sql; create table det_unadj (drop=permno_date repweek) 

as select * 

from det_unadj as a left join crspadj (rename=(CFACSHR=crspadj_rep date=permno_date)) as b 

on a.permno=b.permno and a.repweek=b.permno_date; 

quit; 

 

title 'Merged Detail History with CRSP adjustement factors'; 

proc print data=det_unadj (obs=50 keep=ticker fpedats estdats analys fpi forecast forecast_adj   

                           repdats actval_unadj actval_adj crspadj_est crspadj_rep); 

run; 

 

proc sort data=det_unadj;  

 by ticker date usfirm; 

run; 

 

/* -------------------------------------------------------- */ 

/* Merge the Adjustment Factors into the Detail History     */ 

/* Actual data and calculate the unadjusted estimate values */ 

/* -------------------------------------------------------- */ 

data ibestemp; 

merge det_unadj (in=detdata)  

adjfac (in=adjdata); 

by ticker date usfirm; 

retain lcadjfac lspdates; 

format CADJFAC 12.6 SPDATES date9.; 

if first.ticker or adjdata then do; 

  lcadjfac = cadjfac; 

  lspdates = date; 

end; 

if not adjdata then do; 

  cadjfac = lcadjfac; 

  spdates = lspdates; 

end; 

adjfacbacked=round(forecast/forecast_adj);       /*back out the adjustment factor        */ 

actual_adjfac = actval_adj * cadjfac;           /*method 1 of correcting actual value */ 

actual_adjbacked=actval_adj* adjfacbacked;      /*method 2 of correcting actual value */ 

actual_frcrsp_est=actval_unadj*(crspadj_est/crspadj_rep);/* method 3 of correcting actual value */ 

Label adjfacbacked='Adjustment factor backed out from Adjusted and Unadjusted IBES files'; 

Label actual_adjfac='Actual value corrected using method 1'; 

Label actual_adjbacked='Actual value corrected using method 2'; 

Label actual_frcrsp_est='Actual value corrected using method 3'; 

Label forecast='Analyst forecast (Unadjusted)'; 

Label actval_unadj='Actual value (Unadjusted)'; 

Label crspadj_rep='CRSP Cumulative Adjustment Factor as of Report Date'; 

Label crspadj_est='CRSP Cumulative Adjustment Factor as of Estimate Date'; 

drop date lcadjfac lspdates actval_adj ; 

if detdata then output; 

run; 

 

proc sort data=ibestemp; 

  by ticker fpedats fpi estdats revdats; 

run; 

 

proc sort data=ibestemp; 

  by ticker estdats usfirm; 

run; 

 

/* ----------------------------------- */ 

/* Merge in the identifier information */ 

/* ----------------------------------- */ 

data ids (keep=ticker estdats oftic cusip cname usfirm); 

set ibes.id; 

where &screen; 

label sdates = "Estimate Date, SAS Format"; 

rename sdates=ESTDATS; 

run; 

 

data detu (compress = yes); 

merge ibestemp (in=estdata) ids(in=iddata) ; 

by ticker estdats usfirm; 

retain loftic lcusip lcname; 

if iddata or first.ticker then do; 

  loftic = oftic; 

  lcusip = cusip; 

  lcname = cname; 

end; 

if not iddata then do; 

  oftic = loftic; 

  cusip = lcusip; 

  cname = lcname; 

end; 

if estdata then output; 

drop loftic lcusip lcname; 

run; 

 

proc sort data=detu; 

  by ticker fpedats fpi estdats revdats; 

run; 

 

Title 'Merged Detail History with corrected values of Unadjusted Actuals'; 

Title ' Variable Actual_adjfac obtained via first method'; 

Title ' Variable Actual_adjfacbacked obtained via second method'; 

Title ' Variable Actual_frcrsp obtained via third method'; 

 

proc print data=detu (where=(fpi='1') keep= ticker estdats repdats fpi  

                      forecast actval_unadj actual_adjfac actual_adjbacked actual_frcrsp_est 

                      crspadj_rep crspadj_est obs=100); 

run; 

 

proc download data=detu (keep=ticker fpedats estdats revdats repdats fpi 

analys broker forecast actval_unadj actual_adjfac actual_adjbacked actual_frcrsp_est 

crspadj_rep crspadj_est) out=work.unadjust_merge; 

run; 

endrsubmit; 


